


## Part 1 – The quick method using datasheets

free for use under [Creative Commons 4 Attribution](#) 9 May 2022 ver 1v7


# The VDMOS model parameters

There are many MOSFET's in the LTspice MOSFET library all with subthreshold conduction (since January 2017). Temperature coefficients were added to the VDMOS model May 2019 [ref Help]. But the standard library has not yet been updated to use these temperature coefficients (as of July 2019). Library models do have fixed temp co's, like Bex=-1.5 for Kp, Vtote=-1mV/°C, but are invariably too low for typical power MOSFET's.


This document has been updated (July 2019) for extracting the new temperature coefficients for the VDMOS which includes the body diode.



**Illustration 1:**  
VDMOS + body diode



**Illustration 2:**  
MOSFET caps



**Illustration 3:** IRFP240 Vishay

**Illustration 1** shows the VDMOS and the body diode components.

**Illustration 2** shows the internal MOS capacitors with the MOSFET. These capacitors are with NO MOSFET current ( $V_{gs}=0$ ). In the VDMOS  $C_{gs}$  is approximated as a fixed value.  $C_{gs}$  can't be measured directly due to the other capacitances;  $C_{gs}$  can be found by solving simultaneous equations from measured  $C_{iss}$ ,  $C_{rss}$  and  $C_{oss}$  which are given in datasheets, e.g. **Illustration 3** from [Vishay IRFP240 datasheet](#). Notice  $C_{gd}$  is easy because it is  $C_{rss}$  on datasheets.

Because both  $C_{gd}$  and  $C_{ds}$  are nonlinear capacitors with  $V_{ds}$  (**Illustration 3**) both  $C_{gd}$  and  $C_{ds}$  require a special capacitance model inside the VDMOS model [ref Help]. The VDMOS uses parameters  $C_{gdmax}$  and  $C_{gdmin}$  and swing parameter 'a' for  $C_{gd}$  (same as  $C_{rss}$  in the plots). The source-drain capacitance is modelled by the graded capacitance of the body diode across the source-drain (as shown above with body diode resistance  $R_b$  in series with this diode).

Equations can be used to convert from  $C_{iss}$ ,  $C_{rss}$  and  $C_{oss}$  to  $C_{gs}$ ,  $C_{gd}$  and  $C_{ds}$  in the example below courtesy Hendrik Zwerver [3]. I have found you don't need to use any maths if you guess ball-park values and put them into a plotter jig that shows capacitance directly so you can iterate easily to get the same plots as the datasheet. Several jigs available for download at my site [11].

## Part 1: Using equations to estimate VDMOS parameters from datasheets

This first example below uses datasheet values and equations to get an idea of where to get values and how good these first-time-around values are compared to a finalized model. I chose the IRFP240 because it is in the LTspice standard library.

**Table 1:** Extracting parameters using Hendrik Jan Zwerver's equations [Ref 3]

#	VDMOS parameter	Look for these values on the datasheet
1	Kp Transconductance parameter	= $G_{fs}$ [*ignore units, typ. $G_{fs}$ given at $\sim 1/2 I_{Dcont}@25^{\circ}C$ ]

2	Vto Threshold voltage (at Tamb 25°C)	= Vgs(th) [*assume middle value of max/min range]
3	Rs	= (Vgsmax - $\sqrt{(2 \cdot Id_{max}/Gfs)} - Vto)/Id_{max}$
4	Rd	= Rdson - Rs - 1/[Gfs(10 - Rs/Rdson - Vto)]
5	Ksubthres	= 0.1 [default ]
6	Cgdmax	=Crssmax+(Crssmax-Crssmin)/ATAN(10000)
7	Cdgmin	=Crssmin
8	A (nonlinear Cgd slope factor at Vds=0)	Range=0.2–2.5, Typical=0.3, Default=1
9	Cgs (assumed constant in the VDMOS)	=Ciss - Crss
10	Cjo body diode Cds @ Vds=0	=(Crssmax-Crssmax*Cgs/(Crssmax+Cgs))
11	M Body diode grading coefficient (Default=0.5)	=-LOG(Cds <sub>25V</sub> /Cjo)/LOG(1+25/0.75)
12	Vj Body diode junction potential (Default=1)	“To be sure set Vj=0.75” [says Hendrik]
13	Rg Gate ohmic resistance	=Rg [*not often quoted, try Rg $\approx$ f-3dB/(2*pi*Cgs), else Rg $\approx$ Ton/(2.2*Ciss) - RgJig]
14	Rb	=Vd/Id -N*Vt/Id*Ln(Id/Is +1) Vt=KT/q=26mV 27°C
15	Is	try Idss [*body diode]
16	N	[*body diode n-ch typ. N=1-2, p-ch typ N=2-6]
17	Pchan	Specify for P-channel within the .model (...) brackets
18	Rds (optional)	[*not normally needed – only add if simulation convergence fails, eg add Rds=1e7 for a 10A MOSFET]
19	Ron (optional)	[manufactures value]
20	Qg (optional)	Qg [max total gate charge Vgs=10V, IDcont]]
21	Vds (optional)	[manufactures absolute Vmax]
22	Mfg (recommended)	=VishMe1907dd [your model date code, eg, YYMMDD]

\* My comments added.

## Step 1: Using Hendrik's estimates for the IRFP240

I'm using the 9 May 2012 datasheet.

**Table 2: Hendrik's approach for the Vishay IRFP240**

#	VDMOS parameter for IRFP240	Datasheet (start at 25C)
1	Kp	= 6.5
2	Vto	= 3
3	Rs	=(10- $\sqrt{(2 \cdot 60/6.5)}-3)/60 = 45m$
4	Rd	= 0.18 - 0.045-1/[6.5(10-45m/180m-3)]=112m
5	Ksubthres	Use 0.1 [default ]
6	Cgdmax	=Crssmax+(Crssmax-Crssmin)/ATAN(10000)=1.3n

7	Cdgmin	= Crssmin = 25p
8	'a' (nonlinear Cgd slope factor at Vds=0)	Typical 0.3 defaults to 1.0
9	Cgs	=Ciss-Crss = 2500p-1300p = 1.2n
10	Cjo	=(Cossmax-Crssmax*Cgs/(Crssmax+Cgs)) = 2400p – 624p = 1780p
11	M	=-LOG(Cds <sub>25v</sub> /Cjo)/LOG(1+25/0.75) = – Log(287p/1780p)/1.536 = 0.52
12	Vj=0.75	To be sure set Vj to 0.75 [says Hendrik]
13	Rg	$Rg \approx \text{Ton}/(2.2 * \text{Ciss}) - Rg_{\text{Jig}} = 51\text{ns}/(2.2 * 1.3\text{nF}) - 9.1 = 9\text{R}$
14	Rb	=Vd/Id – N*Vt/Id*Ln(Id/Is+1) = 1.5/60 – 0.026*Ln(60/25u) = 19m
15	Is	= 25u
16	N=1.5	= 1.5 [body diode n-ch typ. N=1-2, p-ch typ N=2-6]
17	pchan	Don't specify (only if p-chan)
18	Rds	= 0.18
19	Vds	= 200 [datasheet absolute Vmax]
20	Ron	= 180m
21	Qg	= 45nC [Vgs=10V, Id=18A]
22	Mfg	=VishMe1907dd
23	Lambda*	=1/V <sub>A</sub> where V <sub>A</sub> is the Early voltage (x-axis intercept)

\*Lambda was not included by Hendrik. A quick method uses the inverse of the Early voltage from x-axis intercept.

**Summary** for the 'Quick' IRFP240 using Hendrik's approach is below:

	Vto	Kp	Lambda	Rs	Rd	Ksub	Cgd max	Cgd min	a	Cgs	Cjo	M	N	Rb
Quick-v1	3 / 4†	6.5	0*	45m	112m	0.1	1.3n	25p	1*	1.2n	1.8n	0.52	1	19m
Final	4	6	3m	40m	150m	0.15	1.34n	10p	1*	1.2n	1.n	0.4*	1.5	10m

\* unspecified parameters take the default values see LTspice Help file >F1>'M' for MOSFET > scroll down to VDMOS.

† See text. We are free to choose Vto anywhere in the Max to Min range.

The actual model listing is below:


```
*VDMOS with subthreshold (c) Ian Hegglun for 25C
.model IRFP240-v1 VDMOS (Rg=9 Vto=4 Kp=6.5 Rs=45m Rd=112m Ksubthres=0.1
+ Cgdmax=1.3n Cgdmin=25p Cgs=1.2n Cjo=1.8n m=0.52 VJ=0.75 IS=25u N=1.5 Rb=19m
Vds=200 Ron=0.18 Qg=45nC mfg=IH1907)
```

Next I use a plot jig to compare the two models. It can be downloaded [Ref 11](#). How to use [Appendix 1](#).

To compare the Quick model with the Final model (below) and I change the Vto of the Quick model to match my Final model which is 4V. The Vto varies so much from batch to batch so anything in the Max to Min range can be used, in this case anything in the range of 2-4V will do. My Final model has Vto=4 and that was to match the LTspice IRFP240 internal model and 4V matches to the Cordell IRFP240C model. If you want to get the Vto to match a batch of IRFP240's that you have then take a sample and use the mean Vto for your simulations.

My Final model is given below with comparison plots:

```
*VDMOS with subthreshold (c) Ian Hegglun
.model IRFP240h VDMOS (Rg=17 Vto=4.0 Kp=6 Lambda=3m
+ Rs=40m Ksubthres=0.15 Mtriode=0.35 Rd=0.15
+ Bex=-2.3 Vtotc=-6m Tksubthres1=4m Trs1=3.5m Trd1=5m
+ Cgdmax=1.3n Cgdmin=10p a=0.35 Cgs=1.2n Cjo=1n
+ m=0.4 VJ=0.75 IS=10n N=1.5 Eg=1.5 Rb=10m Trb1=2.5m
+ Vds=200 Ron=0.15 Qg=45nC mfg=VishIH1907)
```



**Id vs Vgs (L) and Id vs Vds (R). Two IRFP240 models Red = Quick-v1, Green = Final**

Our Quick-v1 model using Hendrik's approach looks pretty darn good, don't you think?

Now some tweaking to get the drain conductance better – by adding parameter Lambda.

## Step 2: Tweaking the VDMOS parameters

Choosing a rough value for Lambda is required for the **Id vs Vds plot slope**. Lambda affects the Id vs Vgs plot at high currents. A quick method to find Lambda is to use the inverse of the Early voltage from x-axis intercept.

The IRFP240 datasheet Figure 1 **Id vs Vds plot slope** is positive only for Vgs=4.5V and 5.0V but higher than this the slope turns negative which is caused by self heating, so the higher ones cannot be used for finding Lambda.

For the 5.0V plot the current rises from 2.5A to 2.8A for Vds of 20V to 50V.


Slope =  $300\text{mA}/30\text{V} = 0.01 \text{ A/V}$ , then use

$\text{Lambda} = \text{Slope}/\text{Id} = 10\text{m}/2.5 = 4\text{m V}^{-1}$

The model listing for Quick-v2 becomes:

```
*VDMOS with subthreshold (c) Ian Hegglun for 25C
.model IRFP240-v2 VDMOS (Rg=9 Vto=4.0 Kp=6.5 Lambda=4m Rs=45m Rd=112m
Ksubthres=0.1 Cgdmax=1.3n Cgdmin=25p Cgs=1.2n Cjo=1.8n m=0.52 VJ=0.75 IS=25u
N=1.5Rb=19m Vds=200 Ron=0.18 Qg=45nC mfg=IHKT1907)
```

The new plot are shown below. Red = Quick-v2, Green = Final model, Tj is 25°C.




**Id vs Vgs (L) and Id vs Vds (R). Two IRFP240 models Red = Quick-v2, Green = Final**

We now have some slope in the Id vs. Vgs plot. This is probably good enough for 25°C.

Next add Temp Co's for Vto and Kp to look like the 150°C plot (datasheet Figure's 2, 3).

### ***Tweaking the Temp Co's***

The plot below is the same as above (Quick-v2) but now with Temp stepped 25°C and 150°C (as per the datasheet).



**Id vs Vgs (L) and Id vs Vds (R), Step Tj 25°& 150°C. Red = Quick-v2, Green = Final**

The temp. co. for Vto is VtoTc and for Kp is called Bex (Beta exponent where Beta is an alternative name for Kp).


The default VtoTc is -1mV/°C for n-channel (+1mV/°C for p-channel). Typically power MOSFET's are in the 3mV/°C to 10mV/°C range. The default Bex is -1.5 and typically power MOSFET's are in the -2 to -2.5 range.

We can choose mid-range values VtoTc = - 5mV/°C and Bex=-2.2 and see how that looks. See the



plots below.

```
*VDMOS with subthreshold (c) Ian Hegglun fitted to 25C and 150C
.model IRFP240-v3 VDMOS (Rg=9 Vto=4.0 Vt0tc=-5m Kp=6.5 Bex=-2.2 Lambda=4m Rs=45m
Rd=112m Ksubthres=0.1 Cgdmax=1.3n Cgdmin=25p Cgs=1.2n Cjo=1.8n m=0.52 VJ=0.75
IS=25u N=1.5 Rb=19m Vds=200 Ron=0.18 Qg=45nC mfg=IHKT1907)
```




Id vs Vgs (L) and Id vs Vds (R), Step Tj 25°& 150°C. Red = Quick-v3, Green = Final

The new Id vs. Vgs plot now tracks good enough with temperature.

The initial slope of the Id vs Vds plot (near the origin) could be improved to match the datasheet plot by increasing Rd. Once the initial slope is about right then reduce Mtriode to get to turn around at the top to match the datasheets turn around. Increase Rd to 150m and then Mtriode to 0.3 and we get the plots below.

```
*VDMOS with subthreshold (c) Ian Hegglun fitted to 25C and 150C
.model IRFP240-v4 VDMOS (Rg=9 Vto=4.0 Vt0tc=-5m Kp=6.5 Bex=-2.2 Lambda=4m Rs=45m
Rd=140m Ksubthres=0.1 Mtriode=0.3 Cgdmax=1.3n Cgdmin=25p Cgs=1.2n Cjo=1.8n
m=0.52 VJ=0.75 IS=25u N=1.5 Rb=19m Vds=200 Ron=0.18 Qg=45nC mfg=IHKT1907)
```




Id vs Vgs (L) and Id vs Vds (R), Step Tj 25°& 150°C. Red = Quick-v4, Green = Final

The new Id vs. Vds plot now looks good enough. We are about done for the DC parameters. The AC parameters are next then the body diode.

But before doing the AC parameters you may like to compare the Quick-v4 plots with the model already in LTspice:

```
.model IRFP240 VDMOS (Rg=3 Vto=4 Rd=72m Rs=18m Rb=36m Kp=4.9 Lambda=0.03  
Cgdmax=1.34n Cgdmin=0.1n Cgs=1.25n Cjo=1.25n Is=67p ksubthres=0.1  
mfg=International_Rectifier Vds=200 Ron=180m Qg=70n)
```

Note: This model uses default temperature coefficients Bex=-1.5 and Vto has -1mV/°C (for n-ch).



Id vs Vgs (L) and Id vs Vds (R). Red = Quick-v4, Green = Lt-internal lib.

The huge difference could be due to using a different datasheet since the IRFP240 was once available for two different manufacturers with different process settings.

### Getting LTspice to generate curves from datasheets ?

In the above graphs you can easily see how close your Quick models is to my Final model. This is useful for this tutorial. But usually you can't do this. Wouldn't it be nice if we could get LTspice to generate curves from datasheets? Well, it can be done. But it takes about 30 minutes per curve and you want 2 curves (one Id vs Vgs and one Id vs Vds) and this makes getting a Quick model done twice as long. [Appendix 5](#) explains how to use the free Graph Grabber software.


### Tweaking the gate charge

There are two jigs, one for plotting gate charge (Gate voltage vs Gate charge) and another to plot Drain capacitance vs Vds. Start with a=0.3.

First the jig to plot Gate charge [[Ref 11](#)] to check IRFP240 datasheet plot Figure 6.

Parameters Cgs, Cgdmax and Cgdmin and Cjo can be obtained from IRFP240 datasheet plot Figure 5 which is annotated below.





**Illustration 4: IRFP240 Typical Capacitance vs. Drain-to-Source Voltage (DS Fig. 5)**


You should have a go at using this jig to step Cgs, eg,

```
.Step param Cgs List 500p 1000p 1500p
```


Stepping Cgs shows the start *ramp slope* falls with higher Cgs. For the IRFP240 the Cgs was already correct at 1,2nF.

You can see what effect each parameter has by enabling stepping one parameter at a time, eg,

```
.Step param Cgdmax List 1n 2n 3n
```



**Illustration 5. IRFP240 DS Fig.6 Datasheet Gate charge plot**



**Illustration 6. Quick-v4 model (Blue & Red) with Cgdmax stepped 1n, 2n, 3n, Vds=100V.**


Stepping Cgdmax causes the top end slope to decrease as well as shift the break-point from the plateau region (using an idealised intersection of straight lines like the datasheet plots). Therefore choose Cgdmax to get the slope of the datasheet top-end.

The already chosen value for **Cgdmax of 1.3nF** is near enough; notice it is the same as in the datasheet Fig.5. Using Cgdmax from Fig.5 is not always the best value. Use the jig to check the slope and intersection with the plateau. Sometimes you can't find a value that agrees with both Fig.5 and Fig.6 in which case I split the error over the 3 capacitances (Cgdmax, Cgdmin and Cgs).

Cgdmax and Cgdmin values are needed for power amplifier simulations because they introduce high frequency distortion and affect the compensation for clean clip recovery. It is not so important to get accurate values for the model as long as the maximum and minimum Cgd values do vary with

V<sub>ds</sub> – which is why we need the VDMOS model. Most power MOSFET models from manufacturers do not model this effect at all!

Next C<sub>gdmin</sub>. This is C<sub>rss</sub> limit at high V<sub>ds</sub> from the IRFP240 datasheet (Figure 5 Capacitance vs. Drain-to-Source Voltage). It's about 10pF at the limit of the graph which ends at 50V. It might still go lower but use 10pF for now. Run the jig stepping C<sub>gdmin</sub> over say 5p, 10p and 20p to see how these values affect the total gate charge plots. C<sub>gdmin</sub> changes the spacing when V<sub>ds</sub> is stepped without changing the slope.



**Illustration 7: IRFP240 G<sub>gdmin</sub> stepped 5p, 10p, 20p**

Parameter 'a' is then chosen to shift the break point (idealised intersection using straight lines) from the plateau region. Start with a=0.3. Increasing 'a' shifts the break point to the left. We will use **C<sub>gdmin</sub>=10pF and a=0.3**.

Parameter C<sub>jo</sub> is for the body diode. C<sub>jo</sub> is added to the MOSFET's intrinsic C<sub>ds</sub> giving total C<sub>ds</sub>. We only need find a value for C<sub>jo</sub> (not C<sub>ds</sub>) and this is done in the next section for the body diode.

BTW if you are interested in how this jig works – it injects a current ramp of 1nA per second into the gate while a drain current of 18A is forced through the drain (as per the datasheet test) and the drain-source voltage is pulled down as a result of the gate voltage changing. There is Miller feedback which is what causes the plateau region. Plotting V<sub>gs</sub> against time gives the x-axis units of nano-Coulombs (nC) since gate current is nA *per second* and  $Q = \text{Current} \times \text{time}$ .

### ***Tweaking the body diode capacitance***

C<sub>jo</sub> is the diode junction capacitance at V<sub>ds</sub>=0. It is extracted using another jig for plotting total C<sub>ds</sub> versus V<sub>ds</sub> which is the same thing as plotting C<sub>oss</sub> vs. V<sub>ds</sub> [Ref 11].

The C<sub>ds</sub> plot also allows the M parameter for the body diode capacitance, where M changes the rate of fall off of C<sub>ds</sub>/C<sub>oss</sub> with V<sub>ds</sub>.


The first parameter is C<sub>jo</sub>. This can usually be got from the Capacitance vs. Drain-to-Source Voltage (IRFP240 datasheet Fig.5). C<sub>jo</sub> = 2400pF – 1250pF = 1150pF. Hendrik's calculation above gave 1800pF but it seems using Fig.5 method get's the initial value for C<sub>jo</sub> closer. A rounded value of 1000pF (**C<sub>jo</sub>=1n**) will be used for the -v5 model below.

You could step C<sub>jo</sub> in the jig but it doesn't help much. Try

```
.Step param Cjo List 2n 1.5n 1n
```

Next the parameter 'M' using the jig

```
.Step param 'M' List 0.3 0.4 0.5
```




**Illustration 8: IRFP240 body diode 'M' stepped 0.3, 0.4, 0.5**

Probing Cds at Vds=50V is about right with M=0.4.

BTW if you are interested in how this jig works – this jig forces a *ramp voltage* across the drain-source with the gate voltage set to zero, and the drain current is plotted against Vds. With a voltage ramp of 100 volts in 100us we get 1A and so we scale the Id plot by a factor of 1n to read the current as nano Farads. (This only works for 100V in 100us so don't change just the voltage to reset the plot range, but scale both Vpk and sweep time together, eg, 50V in 50us).

Next, parameter 'a'. It determines the slope of Cds decrease as Vds increases due to Cgdmax. It is fitted using the Cds vs. Vds plot jig after Cjo, Vj and M have been chosen. Fit 'a' using the Cds value when Vds=10V as found from the Capacitance vs. Drain-to-Source Voltage (IRFP240 datasheet Fig.5).



**Illustration 9: IRFP240 MOSFET parameter 'a' stepped 0.3, 0.5, 1, 2**


Probing Cds at Vds=10V is about right with **a=0.3**.

Parameter 'a' only affects the clipping region at high frequencies so is not important for audio amplifier simulations.

It would pay to go back and check the Gate charge to re-tweak Cgdmax and Cgdmin due to the new 'a' value above.

### ***Tweaking gate resistor Rg***

The gate resistor can be tweaked using the Rise time jig (download with the other jigs [Ref 11](#)).



*Illustration 10: IRFP9240 rise time jig with Rg of 9 ohms*

Set the gate voltage to the datasheet test circuit and the external gate and drain resistors to the values in the table that show the rise and fall times. For the IRFP9240 the typical rise time is 43ns with  $V_{DD} = -100V$ ,  $I_D = -11A$ ,  $R_{Gext} = 9.1\Omega$ ,  $R_{Dext} = 8.6\Omega$  and a -10V input step. An Rg of 9 ohms gives 34ns rise time which is close enough for the IRFP9240. The Rg for the IRFP240 was chosen as 17 ohms (was 9 ohms from Hendrik's equation).

### **Summarising so far:**

The new AC parameters in the -v5 model are below:

```
*VDMOS with subthreshold (c) Ian Hegglun fitted to 25C and 150C
.model IRFP240-v5 VDMOS (Rg=17 Vto=4.0 Vt0tc=-5m Kp=6.5 Bex=-2.2 Lambda=4m
Rs=45m Rd=140m Ksubthres=0.1 Mtriode=0.3 Cgdmax=1.3n Cgdmin=10p a=0.3 Cgs=1.2n
Cjo=1n m=0.4 VJ=0.75 IS=25u N=1.5 Rb=19m Vds=200 Ron=0.18 Qg=45nC mfg=IHKT1907)
```

### ***Tweaking the body diode DC parameters***

Use the body diode jig for DC parameters [\[Ref 11\]](#). These parameters are not significant for audio power amplifier simulations – they only affect reverse conduction if the SOA trips and audio SSR's. Body diodes models are important for resonant SMPS's.




Illustration 11 IRFP240 body diode




Illustration 12 IRFP240 diode model

The first two parameters to fit are the diode saturation current parameter 'Is' and small Id slope 'N'. Slope 'N' is extracted at 25°C. It has temp co parameters 'Eg' and 'Xti' for fitting the 150°C plot. With N=1 the slope is 60mV per decade of current at 25°C. The datasheet slope is about 1 decade per 100mV which means N is around 1.6. I chose **N=1.5** which is close enough.

The value of 'Is' was chosen by starting at 1pA and working up using the plot jig. Increasing 'Is' by a factor of 10 roughly doubles the current ( $\ln_e(10) = 2.3$ ). Fit 'Is' using the low voltage end of the datasheet plot. The lowest for the IRFP240 datasheet is 0.7V for which Id is -0.6A for 25°C (diode current is technically negative for the n-channels since it is reverse drain current). **Is=10n** is OK.

Next choose Rb (body diode series resistance) for 25°C. This determines the high current region. If the current is too high then increase Rb. A quick starting value can be found (more easily than Hendrik's equation) by finding the rough change in current and change in voltage at the top end of the Id vs. Vd plot; for the IRFP240 datasheet at 25°C this is 10A for 0.1V, so try **Rb ≈ 0.01 ohms**.

This estimated Rb value was found to be about right using the plot jig. So it's **Rb=10m** at 25°C.

Next for 150°C two parameters are used: 'Trb1' for the high current end and 'Eg' for the low current end. 'Eg' is the band-Energy gap voltage which determines how temperature changes 'Is'. Parameter 'Eg' by default is 1.11V. In our example with N of 1.5 we can try 'Eg' = 1.5 for a starting value.

**Eg=1.1** was found to a good fit for 150°C. Parameter Trb1=2.5m is chosen to get the crossing current to match the datasheet plot (crossing at 40A for the IRFP240).


Now we can list the new body diode parameters as the -v6 model below:

```
*VDMOS with subthreshold (c) Ian Hegglun fitted to 25C and 150C
.model IRFP240-v6 VDMOS (Rg=9 Vto=4.0 Vtotc=-5m Kp=6.5 Bex=-2.2 Lambda=4m Rs=45m
Rd=140m Ksubthres=0.1 Mtriode=0.3 Cgdmax=1.3n Cgdmin=10p a=0.3 Cgs=1.2n Cjo=1n
m=0.4 VJ=0.75 IS=10n N=1.5 Eg=1.1 Rb=10m Trb1=2.5m Vds=200 Ron=0.18 Qg=45nC
mfg=IHKT1907)
```


We are now done for the IRFP240 model. ... The IRFP9240 is given below:

```
*VDMOS with subthreshold (c) Ian Hegglun fitted to 25C and 150C
.model IRFP9240h VDMOS (Nchan Rg=6 Vto=+3.76 Kp=9 Lambda=4m Rs=64m
Ksubthres=0.15 Mtriode=0.2 Rd=0.25 Rb=0.05 Bex=-2.3 Vtotc=-6m Tksubthres1=4m
Trs1=3.5m Trd1=5m Trb1=0 Cgdmax=1600p Cgdmin=30p a=0.5 Cgs=1400p Cjo=1000p m=0.4
Vj=0.75 N=7 Is=10u Eg=3.15 Vds=-200 Ron=0.5 Qg=44nC mfg=VishIH1907)
```

Notice the IRFP9240 requires a high 'N' value of 7. But in this case 'Eg' of 3.15 is not as high as 'N' like for n-channels. The IRFP9240 body diode plots are shown below:



**Illustration 13 IRFP9240 body diode**



**Illustration 14 IRFP9240 diode model**

### Summarising the finalised Quick model after tweaking:

The new AC parameters in the -v5 model are below:

```
*VDMOS with subthreshold (c) Ian Hegglun fitted to 25C and 150C
.model IRFP240-v6 VDMOS (Rg=17 Vto=4.0 Vt0tc=-5m Kp=6.5 Bex=-2.2
Lambda=4m Rs=45m Rd=140m Ksubthres=0.1 Mtriode=0.3 Cgdmax=1.3n
Cgdmin=10p a=0.3 Cgs=1.2n Cjo=1n m=0.4 VJ=0.75 IS=10n N=1.5 Eg=1.1
Rb=10m Trb1=2.5m Vds=200 Ron=0.18 Qg=45nC mfg=IHKT1907)
```

### For an adventure! Using Graph grabber software to generate Tables for LTspice


Graph grabber software can be used to get many (x,y) points into a spreadsheet ([Appendix 5](#)). Then you can export the data in pairs for use in LTspice Table (...) function to generate both Id vs Vgs and Id vs Vds plots ([Appendix 3](#)).

Measured data for the RFP240 and IRFP9240 was imported into LTspice using Graph Grabber. The plots are shown below after the threshold voltages have been offset to match the Final models obtained from Vishay datasheets. Also the measured plots are scaled by a factor to get the curves to match the Final models. This assumes the Final models are the correct ones and the measured ones are different due to process variations or they are from a different source than Vishay.

The scale factor used for the two measured IRFP240's is 1.1 and the scale factor used for the two measured IRFP9240's is 1.3. This difference in Kp and gm is within the Max and Min range sometimes specified on datasheets. Those that have specified the Max and Min Gm values have about a  $\pm$ variation from the typical value. The typical values are the ones we usually use for our models – unless we have enough samples (at least 10 per batch) over a long time frame (several years) to confidently use our measured means as the typical values.

Notice the p-ch measured data appears to show thermal runaway, possibly not fast enough sweeps with too high Vds. The n-channel drops at the top end and this can be caused when the drain voltage is not high enough to stay in the current saturation region (unlike the reference model which has adequate drain voltage).





*Illustration 15: Gm plots for measured IRFP240/9240 against our Final models*

## Summary

Hendrick's quick method gave quite good estimates for the VDMOS from the datasheets. They could be used without tweaking. But to get good capacitance and body diode parameters with temperature some iterations are needed. And the plotter jigs are indispensable, making tweaking easy.

Once you have worked through this procedure it becomes quite easy. There is no need to import datasheet curves using Grap Grabber software – just a few key points from the datasheet plots are enough. The amount of time spent capturing the data, converting it and plotting it, does not get you that much more fitting accuracy and doubles the time to make a model.

## Acknowledgements

Also thanks to '[Keantoken](#)' (Anthony) a member of the [DIYaudio.com](#) community who motivated me to get all these models done properly. Anthony started the "[better MOSFET...](#)" thread which is a good place to find VDMOS models for audio power amp simulations.

Thanks to [Bob Cordell](#) for asking me to write this paper on how to extract VDMOS models and for referencing this paper in [his book](#) (2<sup>nd</sup> Ed. Ch. 24).

I would like to thank Anthony Holton of [Holton Precision Audio](#) for a pair of Exicon 20N20 and

20P20's samples to bench test for this report.

Thanks to Jan Didden of [Linear Audio](#) for taking a keen interest in my projects.

Thanks again to Steven Benbow for the free GraphGrabber software that can be downloaded [here](#).

Last but not least Mike Engelhardt for making LTspice what it is, and for it's free use. Mike, thanks again for adding those temp co's to the VDMOS May 2019! They make the VDMOS much more accurate over temperature and easier to get then working as they should.

## References

1. Shickman Hodges, Level 1 model, IEEE 1968 paper <https://paklaunchsite.jimdo.com/>
2. Mike Engelhardt, 'Switcher CAD manual scad3', Feb 2010. [Download](#). The Help file in the latest LTspice is usually fully up to date covering recent changes to the VDMOS etc.
3. Hendrik Jan Zwerver, 'LTspice built in VDMOS model', 04 Dec 2006 [http://www.magma.ca/~legg/SR5/LTspice\\_built\\_in\\_VDMOS\\_model.pdf](http://www.magma.ca/~legg/SR5/LTspice_built_in_VDMOS_model.pdf)
4. LTspice 'ChangeLog.txt' entry "12/16/14 Revised the subthreshold conduction equations for VDMOS." This file is located in the 'LTC' directory (with the program exe).
5. Bob Cordell, Book, '[Designing Audio Power Amplifiers](#)', 2<sup>nd</sup> Edition includes new information on the VDMOS subthreshold parameter and more.
6. Francisco J. García-Sánchez; Adelmo Ortiz-Conde; et al, 'A unified look at the use of successive differentiation and integration in MOSFET model parameter extraction', J. Microelectronics Reliability, 2014. Available at [Research Gate](#).
7. F. J. García-Sánchez; Adelmo Ortiz-Conde; et al, 'Revisiting MOSFET threshold voltage extraction methods', J. Microelectronics Reliability, 2012. Available at [Research Gate](#). See Fig. 8, Fig. 9, they illustrate measurement 'noise' using the second derivative method.
8. Hegglun, 'Simple pulse jig for power MOSFETs', June 2018, <https://paklaunchsite.jimdo.com/spice-models/>
9. Dr Arto Kolinummi, 'Audio Power Amplifiers: Towards Inherently Linear Amplifiers', Linear Audio Press 2015, pp307. Ph.D Thesis at Tampere University of Technology, Finland, June 2010. <https://linearaudio.net/books/2220>
10. VDMOS models are posted at diyAudio.com here [VDMOS models](#). Post #1 has fan-out links to the latest versions. Keantoken, thanks for maintaining this resource.
11. Several plotter jigs that are useful for tweaking parameters in this paper can be downloaded here <https://paklaunchsite.jimdo.com/spice-models/>
12. Hegglun, 'A subcircuit to add quasi-saturation to the VDMOS', June 2018, <https://paklaunchsite.jimdo.com/spice-models/>
13. Exicon datasheets <http://www.exicon.info/PDFs/ECW20N20-Z.pdf>  
<http://www.exicon.info/PDFs/ECW20P20-Z.pdf>
14. Kevin Aylward, SuperSPICE , <https://www.anasoft.co.uk/download.htm> includes the LTspice VDMOS model with quasi-saturation (since 2017).
15. Ian Hegglun, 'Hotter Spice', Electronics World, May 1999 p394-9. Part 2 July 1999 p585-9. 2015 updated version <http://paklaunchsite.jimdo.com/pak-downloads/more-downloads/>
16. IRFP240 datasheet <https://www.vishay.com/docs/91210/91210.pdf>  
IRFP9240 datasheet <https://www.vishay.com/docs/91239/sihfp924.pdf>  
IRF640 datasheet <http://www.vishay.com/docs/91036/sihf640.pdf>

## Appendix 1. The LTspice jig for comparing two models

This jig plots two MOSFET's at once so you can see how close your model is to a reference model.

The reference model could be one obtained from the web as a subcircuit that is not in the VDMOS form and you want to convert it to the VDMOS form. To get a subcircuit to run you need to set the Prefix to letter 'X' (Ctrl+Rt click on the MOSFET symbol).

If you don't have a reference model then you could import data from a datasheet. But that takes a lot of time so you can just plot one MOSFET (Delete the unused plot or turn the other off in the plot window).

One very useful feature of this jig is the very easy way to switch from an (Id vs. Vgs) plot to an (Id vs. Vds) plot with only one change to the .dc card/lines (Step 2 below).

The jig can be downloaded from [Ref 11](#). It comes with the IRFP240 and IRFP9240 models (as used for Part 1 of this paper).

### Steps to use the jig

1. Open the file.

2. For Id vs Vgs

Choose the plot type (Top left corner)

1<sup>st</sup> card .dc V1 2 12 0.01

2<sup>nd</sup> card ;dc V2 0 30 0.01 Note the semicolon is present.

Run

Edit .Step param Vgs List 4 6 8 to set the Vgs plot range

Edit . .param Vds=10 to set the Vds to match the datasheet (eg if you have Lambda>0)




Figure A1.1. The jig used to plot two VDMOS models for a comparison

3. For Id vs Vds

Choose the plot type (Top left corner)

1<sup>st</sup> card `.dc V1 2 12 0.01`

2<sup>nd</sup> card `.dc V2 0 30 0.01` Edit out the semicolon and replace it with a full-stop.

Run. At the prompt choose the second card.

Edit `.Step param Vgs List 4 6 8` to set the Vgs plot range.

#### 4. Select one reference VDMOS

Right click on the model you want and select 'SPICE directive' button

Right click on the earlier model (in black) and select 'Comment' button. Run

#### 5. Select a 2nd VDMOS

Right click on the model you want and select 'SPICE directive' button

Right click on the earlier model (in black) and select 'Comment' button. Run

6. Plots can be turned off temporarily using eg `I(B2)*0` and turning on again using `I(B2)*1`.


`I(B1)` plot is for  $I_d$  vs  $V_{gs}$  plots from Table (...) data, and

`I(B2)` & `I(B3)` plots are for  $I_d$  vs  $V_{ds}$  plots from Table (...) data.

7. You can add other models. Other models can also be added as a test file using

`.include <Mymodel.txt>` to the jig (circuit). Place the text file `<Mymodel.txt>` in the same directory as the jig files.

The plot below shows the IRFP240 and IRFP9240 VDMOS models compared to measured data. The gm's are plotted here using the “`d(Id...)`” operator (first derivative of  $I_d$ ). Plotting Gm's gives more useful information about how well the model represents the measured values, after all it is Gm's (gain) and  $I_d$ 's that determine what distortion is generated in an amplifier.



**Illustration 16: Gm's for IRFP240 and IRFP9240 VDMOS models compared to measured data**


Note: The factors used for scaling in the above plots can't tell us much since in this case the data provided did not include the temperature used not the  $V_{ds}$  (since  $V_{ds}$  can changes the effective  $K_p$  value by up to 10% between using 50V or 5V).

## Appendix 2.

### Some equations for extracting Lambda and temp co parameters

#### 1. Extracting Lambda (drain conductance parameter)

The  $I_d$  versus  $V_{ds}$  plots can give an idea of what value to use for Lambda in the VDMOS. Unfortunately, lateral power MOSFET's includes a quasi-saturation region up to a  $V_{ds}$  of 20V as seen in **Figure A2.2** (near the end of the triode region) and this makes it hard to extract Lambda. At low  $V_{ds} < 20V$  the slope is measurable so Lambda can be calculated, but for higher  $V_{ds} > 20V$  the slope is very close to zero.



**Figure A2.2.** Output characteristic curves for [Exicon ECW20N20-Z \[L\]](#) and [ECW20P20-Z \[R\]](#)

One way is to take an average slope to cover the normal operating range of 10V to 100V (eg a 100W/8R amp). For example, using the N-ch  $V_{GS}=3V$  curve there is approximately  $\frac{1}{4}$  A increase from  $V_{ds}=10V$  to 60V which give  $\Lambda=0.25A/50V$  or  $5m[V^{-1}]$ . This is the value used. BTW A simple curve tracer that shows the quasi-saturation effect in the 'Hotter Spice' article [[Ref 15](#)].

#### 2. Extracting the Bex Temperature Coefficient for Kp

For a wide temperature range we use the power-law 'Bex' (Beta exponent):

$$Kp_{Tj2} = Kp_{Tj1} \left( \frac{Tj2 + 273}{Tj1 + 273} \right)^{Bex}$$

where Bex is negative and typically in the range of -1.5 to -2.5. The minus means Kp decreases with increasing temperature, and

where  $Tj1$  is the lower temperature used on the datasheets, usually 25°C. The default temp for SPICE is 27°C so you need to add the statement `.Temp 25` to select the lower temperature for your model set up. To step two temperatures like 25°C and 150°C for datasheets then you can use use:

`.Temp 25 50 .`

An alternative temperature coefficient for Kp is  $J_\beta$  and is calculated using:

$$Kp_{Tj2} = \frac{Kp_{Tj1}}{1 + J_{\beta}(Tj1 - Tj2)} \quad (1)$$

$$\text{so } J_{\beta} = \frac{\left( \frac{Kp_{Tj2}}{Kp_{Tj1}} - 1 \right)}{Tj2 - Tj1} \quad (2)$$

But this formula is not suitable for very low Tj2 (when Tj2 << Tj1) since the denominator can go to zero. For wide temperature range high and low we use the power-law form with exponent Bex.

To converting from J<sub>β</sub> to Bex it was found that a J<sub>β</sub> of 9m/C requires a Bex of -2.5 and a J<sub>β</sub> of zero requires a Bex of zero. So the scale factor for is Bex is -2.5/9m = -273 times (add the minus).

$$Bex \approx -273 \cdot J_{\beta}$$

### 3. The threshold voltage temperature coefficient (VtoTc)

The threshold voltage temperature coefficient VtoTc is used to calculate Vth at a different temperature to Tref:

$$Vth_{T2} = Vto - VtoTc(T2 - Tref) \quad (3)$$

Where Vto is the threshold voltage at the reference temperature Tref (default is 27°C and datasheets is 25°C)

#### Choosing the threshold voltage temperature coefficient using the 'ztc' point

In amplifier simulations we sometimes want the zero temperature coefficient point to be accurate. The ZTC voltage and current can be found from pulse tests or ball park figures from datasheets.

The ZTC point is where the drain current does not alter with changes in junction temperature. It is the result of the rise in drain current from the threshold voltage falling with Tj increasing being cancelled by the fall in Kp (and fall in gm at a particular current) with Tj increasing. Since Id<sub>ztc</sub>(Vgs) is independent of temperature it is possible to find solve for Tcv once we have found Kp's temperature coefficient using:

$$Id_{Tref} = \frac{Kp_{Tref} (Vgs_{internal} - Vto)^2}{1 + J_{\beta}(Tref - Tref)} = Id_{T2} = \frac{Kp_{Tref} (Vgs_{internal} - [Vto - VtoTc(T2 - Tref)])^2}{1 + J_{\beta}(T2 - Tref)}$$

Therefore a helpful initial threshold temperature coefficient is:

$$VtoTc \approx \frac{J_{\beta}}{2} (Vgs_{ztc} - Vto) \quad (4)$$

where J<sub>β</sub> is the temperature coefficient of Kp and the factor of ½ comes from approximating 1-√(1+x) ≈ ½x. We also neglect small differences in Vto's due to Rs. This gives an approximate temperature coefficient. BTW the gm versus Vgs ztc point occurs at a different Vgs to the Id versus Vgs ztc point so don't mistake it for the Id ZTC point.






Figure A2.1. Right datasheet Exicon ECW20N20-Z at low Id. Left 20P20 [Ref 13].

### Appendix 3. How to export points from spreadsheet to LTspice Table(...)

#### Step 1: Set up a new sheet to collate selected points ready for exporting to text file

In a new sheet use equations to get selected (x,y) points ready for exporting to text file. You can get away with plotting in  $V_{gs}$  increments of 100mV for the first 1V above the threshold voltage and then 0.5V steps which requires 30 pairs to cover a 10V  $V_{gs}$  range. Use the equation “=Sheet1:A3” to get the value of cell A3 on Sheet1 into the destination cell in say Sheet2. You do this using the mouse by typing equals in the destination cell of Sheet2 then click on Sheet1 tab and click on cell A3 and press Enter – you are now in sheet 2 with “=Sheet1:A3” automatically entered into the destination cell. Repeat for the other value of the pair. Select the 2 destination cells and drag down and you will see all the other values from Sheet1 in Sheet2. Then remove unwanted ones. In OpenOffice use “Shift+minus>Enter” to delete a row where the cursor is placed (or several selected rows) more quickly.

You can export several different groups at once by adding more groups under the first group. For example  $V_{gs}, I_d$  for 25°C as the first group, and  $V_{gs}, I_d$  for 75°C for the second group,  $V_{gs}, g_m$  for 25°C as the third group, and  $V_{gs}, g_m$  for 75°C as the third group. If you want to include other values such as for the complement FET then add them more groups. If you want to also include datasheet values then add them more groups. Putting all of them on one sheet reduces the amount of post editing next for of the exported text file. In OpenOffice to export first save your spreadsheet, then use SaveAs>csv>remove the " text delimiter and leave blank>OK, then Close the file. Rename the csv file as a txt file.

In the next section you edit the OpenOffice values exported as a “csv” (comma separated values) file (which is a plain text file where each rows is exported as a line with comma separators ending with a return).

**Step 2:** Open the csv file (renamed as a txt file) using OpenOffice Writer and accept the default import character set (Western Europe ASCII US).

For use in LTspice all values need to be in a long string as pairs  $x_1, y_1, x_2, y_2, \dots$  with no returns. Use Search/Replace to remove the returns and substitute commas. In OpenOffice use \$ as the Search character for a return, and as the Replace character use a comma. You also need to tick Regular Expressions “More Options” box to recognise the \$ as special expression. Then click Replace All, then Save the file. Select the data points and Copy/Paste into an LTspice

Table(V(Vin), <paste-here>).

## How to smooth out glitches in datasheet curves

Some datasheets show a step change in current which can be caused by changing ranges on a meter or changing current sensing resistors. The 2012 ALFET datasheet curves have glitches at  $V_{gs}=4V$  as seen in Figure A3.1. This glitch in current can be removed in LTspice using a correction factor of 1.023 with an equation scaling the Table() function using equation

$$I = \text{If}(U(V(Vin) - 4), 1.023, 1) * \text{Table}(V(Vin) + 50m, 0, .0001, \dots)$$

Vin has 50mV added to offset the datasheets curves so they appear on to of the measured curves.

This is valid since ALFET threshold voltages vary by  $\pm 50mV$  from device-to-device (d-MOSFETs can vary by  $\pm 200mV$ ). ALFET Kp is fairly constant between devices of the same polarity (maybe  $\pm 5$  percent) and across batches to within maybe  $\pm 10$  percent.




Figure A3.1. Original Datasheet curves with 'glitch' at  $V_{gs}=4V$

**Tip:** The gm's in Figure 7 are plotted using  $d(I_d)$ . The gm's of the measured data are plotted using Table(...) values as  $I(B5)/V(1v)$  and  $I(B6)/V(1v)$ . Division by 'V(1v)' is a trick to plot currents as gm's to get units A/V to match  $d(I_d)$  units of A/V (shown as  $\Omega^{-1}$  in LTspice plots).

[Appendix 4](#) is moved to the end of Part 2

## Appendix 5. Using a GraphGrabber software

GraphGrabber can be downloaded [here](#) by Steven Benbow. Thanks Steve! The steps are:

1. If a PDF is supplied zoom to 400% and use the Snapshot Tool and drag over the graph. Alternative: use PrtScrn. Save the clipboard to a gif or jpg.
2. Open GraphGrabber then open the graphic file: File>Open image
3. Zoom out/in so the plot fills the working area, then use Crop to restrict the view to just the plot plus a margin
4. Click 'Set X-axis' and drag a line from the peak to the RHS edge, then enter the X values (refer to your original PDF). If you want logarithm scale click 'Log' (just visible under the Max) then 'OK' (to be certain, check it, Right Click on "X-axis" and mouse over 'Change Scale Type' shows which it is, and can change it to 'Log' if not Log).
5. Click 'Set Y-axis' and drag a line from the peak to the bottom edge, then enter the Y values (lesser value then higher value) in this case the higher is  $-40$  and the lesser needs to be calculated from the indicated values.
6. Before adding a data series zoom in say 200% so the curve extends past the screen (you can use the sliders later to move it while capturing points).
7. Click 'Add Data Series' and I started on the LHS at the curve start. I used 100mV increments over

8V gate voltage range and this is adequate for calculating the derivative in a spreadsheet (needed to plot  $dI_d/dV_{gs}$  versus  $V_{gs}$ ).

8. Click 'To End Data Series'. Then File>Export Graph As Data File (default is CSV).

9. Open a spreadsheet eg OpenOffice or Excel.

10. Open the CSV file and copy the data list to the clipboard. Paste the data into the spreadsheet.

Alternative: Open the .csv file using the spreadsheet File>Open.

11. Select the data and add a X-Y scatter graph.

## Appendix 6. New LT-IV temp models compatible with new VDMOS Temp Co's

My earlier 'Tja' and 'Tjp' VDMOS models have been replaced using the new Temp Co's (May 2019). These models only work in LT-XVII (or later).

I continue to use LT-IV due to older PC's. I use a new alternate models that run in LT-IV but uses the new LT-XVII VDMOS parameters. They use curly bracket equations that are backward compatible with the same parameters as the new LT-XVII VDMOS model parameters.

Those continuing with LT-IV can use my new alternate VDMOS model parameters that were extracted for the new model. Also these alternative models will run in LT-XVII without any editing of the models (they are just more cumbersome with those curly bracket equations).

The earlier '.param Tja=<val>' and '.param Tjp=<val>' statements can still be used but in a different way – you now assign Tjp=<val> after each MOSFET name you want to make different temperatures to the other parts – all others are by default at the global 'ambient' temperature.

The following curly bracket equations are used for Vto, Kp, Rs, Ksubthres, Rd, and Rb as seen in the new alternative model for LT-IV below:

```
*VDMOS with subthreshold (c) Ian Hegglun
.model IRFP240_ VDMOS (Rg=5 Vto={4.0-(6m-1m)*(Temp-25)} Lambda=3m
+ Rs={40m*(1+3.5m*(Temp-25))} Kp={6*((Temp+273)/(25+273))**(2.3-1.5)}
+ Ksubthres={0.15*(1+4m*(temp-25))} Mtriode=0.35 Rd={0.15*(1+5m*(Temp-25))}
+ Cgdmax=1.3n Cgdmin=10p a=0.35 Cgs=1.2n Cjo=1n m=0.4 VJ=0.75
+ Is=10n N=1.5 Eg=1.1 Rb={0.01*(1+2.5m*(Temp-25))}
+ Vds=200 Ron=0.15 Qg=45nC mfg=VishIH1907)
```

The corresponding new LT-XVII model is:

```
*VDMOS with subthreshold (c) Ian Hegglun
.model IRFP240h VDMOS (Rg=5 Vto=4.0 Kp=6 Lambda=3m
+ Rs=40m Ksubthres=0.15 Mtriode=0.35 Rd=0.15
+ Bex=-2.3 Vtotc=-6m Tksubthres1=4m Trs1=3.5m Trd1=5m
+ Cgdmax=1.3n Cgdmin=10p a=0.35 Cgs=1.2n Cjo=1n
+ m=0.4 VJ=0.75 IS=10n N=1.5 Eg=1.1 Rb=10m Trb1=2.5m
+ Vds=200 Ron=0.15 Qg=45nC mfg=VishIH1907)
```

Xti by default is 3. The default is used, so Xti is not specified in most VDMOS models.

### On the the new alternative model for LT-IV

Parameter 'Eg' is the temp co for the body diode 'Is' and curly bracket equations are not needed for 'Is' since 'Eg' is a parameter in the old VDMOS. The value of 'Eg' can be copied across unchanged from the new VDMOS model to the alternative model for LT-IV, likewise 'Is' is copied across.

VtoTc and Bex are inserted into the curly bracket equations below (highlighted values):

```
Vto={4.0-(6m-1m)*(Temp-25)}
Kp={6*((Temp+273)/(25+273))**(2.3-1.5)}
```

These values come from the new VDMOS model and are offset by the internal temp co parameters in the old VDMOS (that are carried across to the new LT-XVII as defaults). This means the alternative model for LT-IV above will also operate perfectly in LT-XVII as well with no changes. The curly bracket equations do the offsetting so you don't need to do any calculations (which may introduce an error ☺) and be careful to preserve the brackets around these internal offsets.

## Appendix 7. The VDMOS parameter Mtriode

A useful undocumented parameter in the VDMOS is the 'Mtriode' parameter which scales the voltage where the MOSFET enters/leaves saturation such as for hard clipping allowing a good model fit for lateral MOSFET's with typically less than 1% average error over a wide current range which is very good for the compact equations used in VDMOS model.

Mtriode is applied to the ohmic region equation and the saturation region is not affected by Mtriode. The equation is:

$$I_{d_{triode}} = Kp (1 + \lambda V_{ds}) \frac{W}{L} \left( Mtriode (V_{gs} - V_{to}) - Mtriode^2 \frac{V_{ds}}{2} \right) V_{ds}$$

The equation was RE'd using a jig (available for download ref [11](#)). The following steps are below:

Zwerver: Linear Region:  $V_{ds} < V_{gs} - V_{th}$

$$I_d = Kp \frac{W}{L} (1 + \lambda \cdot V_{ds}) \left( V_{gs} - V_{to} - \frac{V_{ds}}{2} \right) V_{ds}$$

Set  $\lambda = 0$  for now. When  $V_{ds} = V_{gs} - V_{to}$

$$I_{d_{sat}} = \frac{Kp}{2} V_{ds}^2$$

for  $V_{to}=4$ ,  $V_{gs}=5$   $Kp=1$  then  $I_{d-sat}=0.5A$

$$G_D = \frac{d I_d}{d V_{ds}} = Kp (V_{gs} - V_{to}) - V_{ds}$$

for  $V_{to}=4$ ,  $V_{gs}=5$   $Kp=1$  then  $G_D$  ramps down from  $1A/V$  to  $0$ .

Mtriode scales  $G_D$  initial slope and  $V_{ds}$  for the end of the triode region

$$G_D(Mtriode) = \frac{d I_d}{d V_{ds}} = Kp (V_{gs} - V_{to}) \cdot Mtriode - V_{ds} * ?$$

(reality check  $V_{to}=4$ ,  $V_{gs}=5$   $Kp=1$   $V_{ds}=0$   $Mtriode=0.5$  then  $G_D = 0.5A/V \leftarrow$  agrees with jig)

Now  $V_{ds}$  ends the triode region when  $G_D = 0$  so the previous “?”

$$\times (V_{ds} - sat) = Kp \cdot Mtriode (V_{gs} - V_{to})$$

for  $V_{to}=4$ ,  $V_{gs}=5$   $Kp=1$   $V_{ds}=0$   $Mtriode=0.5$  then we want  $V_{ds-sat}$  to be  $2V$  so the “?” factor must be  $Mtriode^2$ , giving

$$G_D(Mtriode) = Kp (V_{gs} - V_{to}) Mtriode - V_{ds} \cdot Mtriode^2$$

To get back to the original equation integrate  $G_D$  after rearranging

$$G_D(Mtriode) = Kp \cdot Mtriode [(V_{gs} - V_{to}) - V_{ds} \cdot Mtriode]$$

integrating gives

$$I_{d_{triode}} = Kp Mtriode \frac{W}{L} (1 + \lambda \cdot V_{ds}) \left( V_{gs} - V_{to} - Mtriode \frac{V_{ds}}{2} \right) V_{ds}$$

use I(B1) to generates the current for the above equation and plot d(I(B1)) with MOSFET d(Id(M1)).

The equation in a behavioural source gives the same as the VDMOS model when Mtriode is changed to whatever. It agrees! So the formula above is correct for Mtriode.

-----oo00oo-----

... Part 2

## Part 2: A direct extraction method for parameters Kp and Vto

free for use under [Creative Commons 4 Attribution](#) 9 May 2022 ver 1v1



**6. Add Lambda, Mtriode and Rd** and compare to measured data


**7. Use LTspice to fine tune the VDMOS parameters.** Check the ZTC current is close to the measured data ZTC point and/or the datasheet value

Add the model definition to your library file. Then simulate Class-AB crossover distortion.


### 1. Extract Kp using a $d\sqrt{I_d}$ versus $V_{gs}$ plot

A  $d\sqrt{I_d}$  versus  $V_{gs}$  plot is shown in **Figure 1-4**.  $d\sqrt{I_d}$  versus  $V_{gs}$  plot is the derivative of the squareroot of drain current versus  $V_{gs}$  plot. The Kp is related to the peak by:

$$Kp \simeq 2(\text{peak})^2 \quad \text{where 'peak' has units of } \sqrt{A/V} \quad (5)$$



**Figures 1. 10N20 measured  $d\sqrt{I_d}$  versus  $V_{gs}$  plots at 25°C (Blue) and 75°C (Orange)**



**Figures 2. 10P20 measured  $d\sqrt{I_d}$  versus  $V_{gs}$  plots at 25°C and 75°C**

We read off the Kp from the peak and we can estimate the subthreshold slope from the rising edge. The peak slope does not give the true Kp because  $R_s$  reduces the peak. [Appendix 4](#) illustrates the effect of  $R_s$  on reducing the peak. We can undo the effect of  $R_s$  to find the true peak value by extrapolating the tail slope back to the threshold voltage intercept, then doubling the increase.

But to do this we need to know  $V_{to}$  which hasn't been extracted yet. So use an approximate  $V_{to}$ , found at 60% of the plot peak (circled points in Figs 1,2).


**Table 1.1 Summary of extracted parameters so far**

MOSFET	$K_p$ at 25C [A/V <sup>2</sup> ]
10N20	1.401
10P20	0.950
20N20	2.454
20P20	1.768


A  $d\sqrt{I_d}$  versus  $V_{gs}$  plot gives reasonably good estimates for the first parameter which is always the hardest to extract and all the other parameters are affected by the first one extracted. Having found the first parameter  $K_p$  the other parameters  $V_{to}$  and  $R_s$  are easier to extract from a  $\sqrt{I_d}$  versus  $V_{gs}$  plot which does not involve any current differencing and is therefore not upset much by measurement noise. There is very little measurement noise **Figure 1**.

Taking the differences of the square-root of  $I_d$  effectively gives us a second derivative without adding any noise rather than methods using the second derivative [Ref 6,7].

The second feature here is the jig minimises measurement noise by toggling a small and accurate gate voltage offset above and below each  $I_d$  point [Ref 8].



**Figures 3. 20N20 measured  $d\sqrt{I_d}$  versus  $V_{gs}$  plots**



**Figure 4. 20P20 measured  $d\sqrt{I_d}$  versus  $V_{gs}$  plots**

For the 20N20 (**Figure 3**) the 25°C plot peaks at  $0.98\sqrt{A/V}$  giving an estimate for  $K_p$  of  $1.9A/V^2$ . But when we use this value in the VDMOS we find that it is about 20% too low. As mentioned above, to find the true peak value by extrapolating the tail slope back to the threshold voltage intercept, then doubling the increase. We get rough estimate from the  $V_{gs}$  where the value is 60% of the peak (circled areas in Figs 1-4).


## 2. Extract $V_{to}$ from the x-intercept using one $\sqrt{I_d}, V_{gs}$ point at the $d\sqrt{I_d}$ vs. $V_{gs}$ peak

Use **Equation 2** to calculate the x-axis intercept:

$$V_{to} \simeq V_{gs_{Peak}} - \frac{\sqrt{I_{d_{Peak}}}}{m_{Peak}} \quad (6)$$

where  $m$  is the peak slope in Step 1,  $V_{gs_{peak}}$  is the voltage for the peak and  $I_{d_{peak}}$  is the  $I_d$  at the peak (at  $V_{gs_{peak}}$ ).

This equation extrapolates back to the x-axis intercept to find  $V_{to}$  from  $V_{gs_{peak}}$  using  $I_{d_{peak}}$  and the slope at  $I_{d_{peak}}$ . This is illustrated by **Figure 5** shows  $V_{to}$  by extrapolating back from the inflexion to the x-axis.



**Figure 5. Extracting  $V_{to}$  from the x-intercepts (20N20 Right, 20P20 Left)**

So we don't need to plot  $\sqrt{I_d}$  versus  $V_{gs}$  – it is only shown here to illustrate Equation 2.

For example the 20N20 (Fig.3) at 25°C with  $V_{gs_{peak}}$  of 0.50V and  $I_{d_{peak}}$  of 134mA and a peak slope

of  $1.06\sqrt{\text{A/V}}$ . Using Equation 2  $V_{to} = 0.5 - \sqrt{0.134 / 1.06} = 155\text{mV}$ .

**Table 2.1 Summary of extracted parameters so far**

MOSFET	Kp at 25C [A/V <sup>2</sup> ]	Vto <sub>25</sub> [V]
10N20	1.401	174m
10P20	0.950	-540m
20N20	2.454	155m
20P20	1.768	-607m

At the end you can see the final value for  $V_{to}$  is the same as this – so it doesn't need any further tweaking! If you had used the standard approach and plotted Figure 5 to get the x-axis intercept then you would get 120mV for the 20N20 at 25C – because you are using the indicated slope at the inflexion ( $0.98\sqrt{\text{A/V}}$ ) rather than the extrapolated slope ( $1.06\sqrt{\text{A/V}}$ ). The standard approach needs several iterations to reach a final value for  $V_{to}$ .

### 3. Calculate Rs in the MOSFET Id equation using Kp & Vto

Use Equation 3 to calculate Rs

$$R_s = \frac{\left( \frac{1.1 K_p}{2 I_{d_{sat}}} (V_{gs_{sat}} - V_{to})^2 - 1 \right)}{1.1 K_p (V_{gs_{sat}} - V_{to})} \quad (7)$$

Where Kp was found in Step 1 and Vto in Step 2 and choose one  $I_{d_{sat}}$ ,  $V_{gs_{sat}}$  point carefully in the 3A to 7A range or 2-3 volts above  $V_{to}$  and while still in saturation. Use the  $d\sqrt{I_d}$  versus  $V_{gs}$  plot to find the highest  $I_{d_{sat}}$ ,  $V_{gs_{sat}}$  point that is part of the general trend.

Leaving saturation causes a sudden dip at high  $V_{gs}$  so choose a point just before that dip (if it is seen). For the 20x20 the current limit where the MOSFET changes from saturation to the linear/triode region is 7A because of the test jig's low  $V_{ds}$  (12V) and a series resistor (0.25Ω effectively added to the drain resistance). Also, the MOSFET's change from saturation to the triode region can be spotted as the gm peak in a gm versus  $V_{gs}$  plot.

In this example the value calculated for  $R_s$  was 121mΩ at  $V_{gs}=3.5\text{V}$  where the gm peaks (end of the saturation region). The best manual fit with  $K_p=2.40\text{A/V}^2$  and  $V_{to}=155\text{mV}$  gives  $R_s=120\text{mΩ}$ . Again, final tweaking did not change the initial extracted parameter value.

**Table 3.1 Summary of extracted parameters so far**

MOSFET	Kp at 25°C [A/V <sup>2</sup> ]	Vto <sub>25</sub> [V]	Rs [Ω]
10N20	1.401	174m	251m
10P20	0.950	-540m	307m
20N20	2.454	155m	121m
20P20	1.768	-607m	151m

### 4. Extract subthreshold slope (SS) for ksubthres using either Log10(Id) vs. Vgs, OR use the $d\sqrt{I_d}$ vs. $V_{gs}$ plot.

Use Equation 4 to calculate the  $k_{subthres}$  parameter value (for subthreshold conduction)

$$k_{subthres} \simeq \frac{1}{SS} \quad (8)$$

The value of the VDMOS subthreshold parameter  $k_{subthres}$  is calculated from is the inverse of the

subthreshold slope (SS) in Volts per decade [footnote <sup>2</sup>].

You can either use the standard  $\text{Log}_{10}(I_d)$  vs.  $V_{gs}$  plot slope OR the  $d(\sqrt{I_d})$  vs.  $V_{gs}$  plot slope. Use equation 5 for the second method

$$\text{Alternative } d(\sqrt{I_d}) \text{ vs } V_{gs}: k_{subthres} \simeq 0.30 \frac{peak}{m_{rise}} \quad (9)$$


where  $m_{rise}$  is steepest part of the rising side of the  $d(\sqrt{I_d})$  vs.  $V_{gs}$  plot (e.g. Figure 1) and 'peak' means the indicated peak of the  $d(\sqrt{I_d})$  curve.

The standard method using the  $\text{Log}_{10}(I_d)$  vs.  $V_{gs}$  plots is shown in **Figure 6** for the measured 20N20 and 20P20 at at 25°C and 75°C.

Notice the 20N20 requires measurements requires a negative gate voltage to -0.2V with very small currents. These currents were measured using a 10 ohm series resistor with the pulse jig with a DMM that could measure down to a few 10 $\mu$ V's.

The alternative  $d\sqrt{I_d}$  vs.  $V_{gs}$  plot method is handy because it does not need very low current measurements; the alternative method for the 20x20 devices uses  $I_d$  currents in the 1mA to 5mA range compared to 10 $\mu$ A-50 $\mu$ A range for the  $\log(I_d)$  method.

The value of  $k_{subthres}$  is not very critical and it can vary by  $\pm 10\%$  without significantly affecting accuracy of distortion simulations.




**Figure 6. Standard method to finding subthreshold slopes (25°C and 75°C)**

The alternative method appears to reduce the error in extracting  $k_{subthres}$  from the parasitic body diode leakage at higher temperature. I found that the 10N20 was different from all the others in the subthreshold region with a significant tail lead-in below a subthreshold  $I_d$  of 300 $\mu$ A in **Figure 7**.

This may explain why the subthreshold slopes for the 10N20 gives  $k_{subthres}$  of 154m and 164m while the alternative method gives only **83m and 95m** for 25°C and 75°C. Typically lateral MOSFET's without this lead-in tail are in the 80m to 120m range – which is the range found above for the other 3 devices without this lead-in tail. This lead-in is probably body diode leakage current.

The subthreshold slopes for the IRF640 and IRF9640 (as IRFP240 and IRFP9240) gives  $k_{subthres}$  of 230m and 200m respectively using the with Equation 4.

<sup>2</sup> In LTspice the slope (in decades per volt) can be read directly by placing the cursors on selected points of a  $\text{Log}_{10}(I_d)$  vs.  $V_{gs}$  plot (after you have typed  $\text{Log}_{10}(I_d)$  as the plot variable, not as a logarithm using the check box).



**Figure 7. 10N20 gives an anomalous subthreshold conduction slope.  $K_{subthres}$  is 95m and 100m for 25°C and 75°C using the alternative method while the standard method gives 154m and 164m about 50% too high, probably due to body diode leakage.**

### 5. Use LTspice to plot the VDMOS model Id

Use the 4 main parameters found above ( $K_p$ ,  $V_{to}$ ,  $R_s$ , &  $K_{subthres}$ ) in LTspice. It is helpful at this stage to import measured data points which can be from a datasheet capture at two temperatures. [Appendix 3](#) covers the steps to set up a Table(...) in LTspice to plot measured values. Plot the Id vs Vgs curves together with the plots generated by a Table(...). You can now proceed to add 3 more DC parameters: Lambda, Mtriode and Rd.

### 6. Add Lambda, Mtriode and Rd and compare to measured data

These are:

- 1)  $\Lambda$  for drain conductance for the Early effect (can be set to 5m initially), and
- 2)  $M_{triode}$  for the triode region and  $R_{dsON}$  and it can be set to 0.5 initially, and
- 3)  $R_d$  which can be set at  $2 \times R_s$  initially.

**Figure 8** shows the Id vs Vgs plots (top 2 plots) are Id for VDMOS models M3 (25°C) and M4 (75°C). The measured values are plotted using I(B3) (25°C) and I(B4) (75°C) using data Table(...).

The lower plots two d(Id) vs Vgs plots are the first derivative of Id which is the gm (forward transconductance). Notice how the VDMOS gm curves *abruptly* from saturation to the triode region, but actual measured curves **are rounded**. There is no parameter in the VDMOS to vary this rounding and this is one reason why the VDMOS model is limited to about 1% when entering the triode region. But don't worry – 1% is good for crossover distortion simulations.






Figure 8. Model Id (Cyan top) and gm (Cyan lower) as extracted (Table 1) and measured Id

Table 1 summarises the extracted values for the 10x20 and 20x20 devices using the above steps and the resulting error. Note this is the error using the *raw values* obtained directly from the spreadsheet and with 'Rd' and 'Mtriode' added manually, they are trimmed in the next section. Appendix \_ gives the equations for Mtriode.

Table 1. Summary of 20x20 and 10x20 parameters as extracted

	$Kp_{25}$ A/V <sup>2</sup>	$Kp_{75}$ A/V <sup>2</sup>	TC 1/K	$Rs_{25}$ $\Omega$	$Rs_{75}$ $\Omega$	TC 1/K	$Vto_{25}$ V	$Vto_{75}$ V	TC V/K	$Ksub_{25}$	$Ksub_{75}$	TC 1/K	Err %*
10N20	1.401	0.990	8.3m	251m	283m	2.6m	174m	108m	-1.3m	83m	95m	2.9m	1.1%
10P20	0.950	0.657	8.9m	307m	366m	3.8m	-540m	-459m	+1.7m	112m	129m	3.1m	2.6%
20N20	2.454	1.792	7.4m	121m	137m	2.6m	155m	76m	-1.6m	90m	90m	0m	1.5%
20P20	1.768	1.247	8.4m	151m	168m	2.3m	-607m	-484m	+2.5m	105m	132m	5.1m	3%

\*Err =  $Abs(I_{d_{meas}} - I_{d_{model}})/I_{d_{meas}}$  then averaged over  $V_{gs}$  0.5V to 7.5V for the n-channel and 1V to 8V for the p-channel.

## 7. Fine-tune the VDMOS parameters for Mtriode and Rd and the ZTC point

Start with fine-tuning the 25C  $I_d$  vs  $V_{gs}$  plot over the first 0.5V after  $V_{to}$  and up to the ZTC point. First trim  $V_{to}$ . Notice the line generated by LTspice for the measured points are interpolations so aim for the *intersection points* and not the lines because in this region of rapid change only the *intersection points* are accurate. My measurements were done in 0.1V steps so there are only about 4 points to fit in this region. Next trim  $K_p$  to align the slope over the next 0.5V above the ZTC point. Then trim  $R_s$  for the best fit over the 1V to 3V above  $V_{to}$ . Above 3V the MOSFET enters the triode region so **trim Mtriode** so the change to triode region occurs at the measured  $g_m$  peak then set  $R_d$  to fit to the tail end region of the  $I_d$  plot.

Table 2 shows the parameters after manual iteration. Figure 9 shows the plots after tweaking.

**Table 2. Summary of 20x20 and 10x20 after manual iteration (including  $\lambda^{\#}$ )**

	$K_{p25}$ A/V <sup>2</sup>	$K_{p75}$ A/V <sup>2</sup>	TC 1/K	$R_{s25}$ $\Omega$	$R_{s75}$ $\Omega$	TC 1/K	$V_{to25}$ V	$V_{to75}$ V	TC V/K	$K_{sub25}$	$K_{sub75}$	TC 1/K	Err %*
10N20	<u>1.30</u>	<u>0.92</u>	8.3m	<u>245m</u>	<u>277m</u>	2.6m	<u>170m</u>	<u>90m</u>	<u>-1.6m</u>	<u>95m</u>	<u>109m</u>	2.9m	<u>0.9%</u>
10P20	<u>0.995</u>	<u>0.745</u>	6.7m	<u>370m</u>	<u>422m</u>	3.4m	-535m	-450m	+1.7m	<u>120m</u>	<u>142m</u>	3.1m	<u>1.2%</u>
20N20	<u>2.40</u>	<u>1.75</u>	7.4m	<u>120m</u>	<u>135m</u>	2.5m	155m	76m	-1.6m	90m	<u>95m</u>	<u>1m</u>	<u>1.2%</u>
20P20	<u>1.85</u>	<u>1.304</u>	8.4m	<u>170m</u>	<u>187m</u>	<u>2.0m</u>	<u>-610m</u>	<u>-500m</u>	<u>+2.2m</u>	105m	130m	5.1m	<u>0.7%</u>

\*Err = Abs( $I_{d_{meas}} - I_{d_{model}}$ )/ $I_{d_{meas}}$  then averaged over  $V_{gs}$  0.5V to 7.5V for the n-channel and 1V to 8V for the p-channel.

† Underlined values indicate a change from Table 1. <sup>#</sup> N-channel  $\lambda=3m$  and p-channel  $\lambda=5m$ .




Figure 9. Tweaked VDMOS  $I_d$  (Cyan) and measured  $I_d$  (Red), see Table 2 for parameters.

Listing 1 provides the final model definition as variable temperature models. See Appendix 6 for an explanation. They can be copied and used in a circuit or put into a user's model text file.

**Listing 1. 10N20 and 10P20 variable temperature models**

```
*VDMOS + subthreshold (c) IanH & keantoken
.model 10N20 VDMOS (Rg=60 Vto=0.17 Kp=1.33 Lambda=3m
+ Rs=0.245 Ksubthres=0.095 Mtriode=0.3 Rd=0.6
+ Bex=-2 Vtotc=-1.6m Tksubthres1=2.9m Trs1=2.6m Trd1=3m
+ Cgdmax=500p Cgdmin=10p a=0.25 Cgs=500p Cjo=300p Trb1=1m
+ m=0.7 VJ=0.75 IS=4n N=1 Eg=1.5 Rb=0.2 Vds=200 Ron=1 mfg=IHK2007)

*VDMOS with subthreshold (c) IanH & keantoken
```

```
.model 10P20 VDMOS (pchan Rg=60 Vto=-0.535 Kp=0.995
+ Rs=0.37 Ksubthres=0.12 Mtriode=0.4 Rd=0.2 Lambda=5m
+ Bex=-2 Vtotc=+1.7m Tksubthres1=3.1m Trs1=3.4m Trd1=0
+ Cgdmax=900p Cgdmin=25p a=0.25 Cgs=900p Cjo=400p
+ m=0.7 VJ=0.75 IS=4u N=2.4 Eg=1 Rb=1 Vds=-200 Ron=1 mfg=IHKT2007)
```

### Listing 2. 20N20 and 20P20 models

```
*VDMOS with subthreshold (c) IanH & keantoken
.model 20N20 VDMOS (Rg=30 Vto=0.155 Kp=2.4 Lambda=3m
+ Rs=0.12 Ksubthres=0.09 Mtriode=0.3 Rd=0.16
+ Bex=-1.9 Vtotc=-1.6m Tksubthres1=1m Trs1=2.5m Trd1=3m
+ Cgdmax=1n Cgdmin=20p a=0.25 Cgs=1n Cjo=1n
+ m=0.7 VJ=0.75 IS=8n N=1 Eg=1.5 Rb=0.1 Vds=200 Ron=0.5 mfg=IHKT2007)

*VDMOS with subthreshold (c) IanH & keantoken
.model 20P20 VDMOS (pchan Rg=30 Vto=-0.61 Kp=1.85 Lambda=5m
+ Rs=0.17 Ksubthres=0.105 Mtriode=0.35 Rd=0.05
+ Bex=-2 Vtotc=+2.2m Tksubthres1=5m Trs1=2m Trd1=0
+ Cgdmax=1.9n Cgdmin=50p a=0.25 Cgs=1.8n Cjo=1n
+ m=0.7 VJ=0.75 IS=8u N=2.4 Eg=2.4 Rb=0.5 Vds=-200 Ron=0.5 mfg=IHKT2007)
```

### Check the ZTC point is close to the measured ZTC point

The Zero Temperature Coefficient (ZTC) intersection point is a convenient reality check.

**Table 3** shows the ZTC values for the measured 20N20 and 20P20.

**Table 3. ZTC values for Datasheet, Measured and VDMOS model**

	10N20		10P20		20N20		20P20	
	Vgs	Id	-Vgs	-Id	Vgs	Id	-Vgs	-Id
<b>Datasheet ZTC</b>	700mV	155mA	1.111V	105mA	700mV	310mA	1.111V	210mA
<b>Measured ZTC</b>	600mV	109mA	1.100V	131mA	790mV	424mA	1.216V	295mA

Notice the datasheet values are significantly different to the measured values indicating the ZTC can vary considerably from batch to batch.

For power amplifier design the ZTC of current is not the best point for the most inherently stable gain linearity and hence most stable distortion with temperature – it is the *ZTC of gm* that is more useful for an inherently temperature stable linearity and distortion.

### ... and compare your model to datasheets

If you have measured a MOSFET and have datasheets available it is useful to compare your model to datasheet curves. You can do this in LTspice by exporting from a spreadsheet the points obtained from the datasheet using a graph grabber ([Appendix 5](#)).

My jig includes 0.25 ohms in series with the drain. To get your model to operate under the same Vds as the datasheets which is Vds=10V (ALFET's) you need to use a 10V supply and no series resistor in the drain (for the power supply's equivalent resistance).

**Figure 10** shows the 20N20 model without the 0.25R sensing resistor and Vds=10V – to give same test conditions as the datasheets.




Figure 10.  $I_d$  using parameters in Table 2 (Cyan) and datasheet  $I_d$  (Red 25°C, Purple 75°C).

The 10N20 and 20N20 datasheet plots align closely with the measured  $I_d$ 's. But the 10P and 20P datasheet plots are significantly higher than measured  $I_d$ 's by a factor of 22%. Maybe the pair I tested was from a non-representative batch? Further measurements are needed on a more recent pair to check if this is a change after the datasheets were compiled.


**8. Add the AC parameters.** See Part 1 for the AC parameters and the body diode DC parameters.

## Simulation Example: Class-AB crossover distortion

Figure 11 shows measured wingspread plots using an AP analyser from [Ref 9] using BUZ901DP and BUZ906DP ( $R_{s_{ext}}=0.15\Omega$ ,  $R_L=8\Omega$ , but  $T_j$  is not specified). The simulated wingspread 20N20-75C and 20P20-75C VDMOS are also shown as Cyan and the EKV is Gold.

The VDMOS generally follows the EKV which matched the measured  $dI_d/dV_{gs}$  versus  $V_{gs}$  plots better than the VDMOS so the EKV is a better guide of actual gm curves (Figure 11). The difference between the EKV and VDMOS at higher currents appears to be due to differences in the VDMOS Lambda and EKV Lambda which uses a very different formulation.

The differences seen between the VDMOS and the EKV at 154mA will affect mainly the lower-order harmonics, probably the 2<sup>nd</sup> to the 7<sup>th</sup> range. This is expected because high-order harmonics are generated by sharp corners of the wingspread plot. Notice the sharp bends are about the same for both the VDMOS and EKV models. Therefore the VDMOS is suitable for simulating Class-AB crossover distortion and is a big improvement on the earlier VDMOS (pre Dec 2014).



**Figure 11. Measured wingspread plots using an Ap analyser from [Ref 10](#) using BUZ901DP and BUZ906DP ( $R_{s_{ext}}=0.15\Omega$ ,  $R_L=8\Omega$ , and  $T_j$  is not specified).**

**Simulated wingspread 20x20-75C for VDMOS (Cyan), EKV (Gold) both at 75°C**

The difference between the EKV/VDMOS versus the measured curves *at higher currents* appears to be due to differences in the EKV and VDMOS Lambda's which cannot be modelled accurately using either the EKV or the VDMOS because the quasi saturation region of audio lateral power MOSFET's is not modelled ([Appendix 2, Fig 2.2](#)).

Otherwise, in the crossover region we can say the VDMOS model is suited for simulating Class-AB crossover distortion and is almost as good as the EKV model. Both the EKV and the VDMOS require additional parameters for accurately modelling power amplifier distortion for low feedback designs; namely, modelling drain conductance below  $V_{ds}$  of 20V to give a smooth transition to the triode region.

As mentioned earlier, this was also a significant limitation when I tried to simulate CMOS inverters in linear mode for guitar soft clipping effects unit using the 4069, 4049 and 74HCU04's using the VDMOS. I also used CMOS inverters in my Cube-law and CSD power amps and developed a subcircuit to better describe the nonlinearity for soft clipping. Contact me if you are interested in my subcircuit to smooth the transition to the triode region. SuperSPICE includes a MOSFET model that is compatible with the LTspice VDMOS and includes quasi-saturation [[Ref 14](#)].

As long as the subthreshold region is smooth, and it is thanks to the *ksubthres* parameter, we can to simulate the high-order harmonics from crossover distortion better, which is the distortion region that we are very sensitive to. Thanks again Mike Engelhardt for adding the *ksubthres* parameter.

===== /▲ \ =====

### **Acknowledgements (repeat from Part 1)**

I would like to thank Anthony Holton of [Holton Precision Audio](#) for a pair of Exicon 20N20 and 20P20's samples to bench test for this report.

Also thanks to 'Keantoken' (Anthony) a member of the [DIYaudio.com](#) community who motivated me to get all these models done properly. Anthony started the "[better MOSFET...](#)" thread which is a good place to find VDMOS models for audio power amp simulations.

Thanks to [Bob Cordell](#) for asking me to write this paper on how to extract VDMOS models and for referencing this paper in [his book](#) (2<sup>nd</sup> Ed. Ch. 24).

Thanks to Jan Didden of [Linear Audio](#) for taking a keen interest in my projects.

Thanks again to Steven Benbow for the free GraphGrabber software that can be downloaded [here](#).



Last but not least Mike Engelhardt for making LTspice what it is and Linear Technology for allowing it's free use. Mike added temp co's to the VDMOS May 2019 and this makes the VDMOS much more accurate over temperature and easier use.

#### Appendix 4. Why Kp is 20% less than the initial slope of gm vs $\sqrt{I_d}$ plots


Subthreshold conduction and source resistance change the initial slope of a  $\sqrt{I_d}$  versus  $V_{gs}$  plot and change the initial slope of a gm versus  $\sqrt{I_d}$  plot. The effect on the  $\sqrt{I_d}$  versus  $V_{gs}$  plot is the extracted Kp is lower and  $V_{to}$  is lower than the values needed in the VDMOS models to generate the exact same curves. The easiest way around this warping effect is to scale up the extracted Kp by some likely amount and then use the corrected Kp to find the new  $V_{to}$  from the corrected slope.

BTW The advantage of using a gm vs  $\sqrt{I_d}$  plot over the  $\sqrt{I_d}$  versus  $V_{gs}$  plot is the gm vs  $\sqrt{I_d}$  plot provides Kp from the initial slope using only one point at the start whereas the  $\sqrt{I_d}$  versus  $V_{gs}$  plot requires two points close to the inflexion point (where the slope is the largest) and this is harder to locate.

Now to answer the question, why the Kp is ~20% lower from the initial slope of a gm vs  $\sqrt{I_d}$  plot, lets look at the effect of first adding  $R_s$  on the  $d(\sqrt{I_d})$  versus  $V_{gs}$  plot and then the effect of adding Subthreshold conduction on the  $d(\sqrt{I_d})$  versus  $V_{gs}$  plot.

**Figure A4.1** shows  $d(\sqrt{I_d})$  versus  $V_{gs}$  plot for 3 cases to illustrate the inversion process:

1. No  $R_s$  and no subthreshold conduction (Black Dotted), and
2. With  $R_s$  and no subthreshold conduction (Gold), and
3. With subthreshold conduction and  $R_s$  (Red)



**Figure A4.1. Comparing peak values of  $d\sqrt{I_d}$  with and without subthreshold conduction. Peak  $d\sqrt{I_d}$  without subthreshold conduction is  $0.7\sqrt{A/V}$  and with subthreshold conduction is  $0.63$ . So Kp with subthreshold conduction can be found by extrapolating of the high  $V_{gs}$  slope back to the threshold voltage, then doubling the increase. Subthreshold slope can be calculated from the rising edge slope.  $R_s$  can be extracted from the high  $V_{gs}$  slope.**

The inversion process can be viewed as a switch function, reaching a constant value on a  $\sqrt{I_d}$  versus  $V_{gs}$  plot, where the switching occurs over about a 1V range either side of the threshold voltage  $V_{th}$  (in Figure A4.1  $V_{th}$  is 0V for convenience and  $K_p$  is  $1A/V^2$ ). Curves for the gm and  $I_d$  curves are included in **Figure A4.1**.

**Case 1** with no  $R_s$  and no subthreshold conduction gives a vertical step in  $d\sqrt{I_d}$  with steady  $1A/V^2$  shown by the solid dotted line ( $0.707\sqrt{A/V}$ ). The gm plot is a straight line from 0V with no  $R_s$  and



no subthreshold conduction and the  $I_d$  plot is a perfect square-law from 0V (not shown). BTW to convert from  $K_p$  to  $d\sqrt{I_d}$  divide  $d\sqrt{I_d}$  by 2 then take the square root (and from  $d\sqrt{I_d}$  to  $K_p$  square  $d\sqrt{I_d}$  then double).

**Case 2** with  $R_s$  added and no subthreshold conduction (Gold curve  $d\sqrt{I_d}$ ) falls off from  $0.707\sqrt{A/V}$  at 0V. The  $g_m$  falls away from the straight dotted line (Cyan). Notice the initial slope of the  $g_m$  versus  $V_{gs}$  plot can give the correct  $K_p$  of 1.0.

**Case 3** with  $R_s$  added and subthreshold conduction (Red curves for  $d\sqrt{I_d}$  and  $g_m$ ) shows the peak is lower than the peak in Case 2 of about  $0.6\sqrt{A/V}$  or an extracted  $K_p=0.8A/V^2$ . Also the slope of the  $g_m$  plot at the maximum (inflexion  $V_{gs} \sim 1.3V$ ) is reduced and this also lowers the x-intercept and so  $V_{to}$  is lower than the value used in the model to generate these curves.

So how can we extract the actual  $K_p$  and  $V_{to}$  values from these curves?

Noting the peak values of  $d\sqrt{I_d}$  with and without subthreshold conduction is  $0.7\sqrt{A/V}$  and with subthreshold conduction is  $0.63\sqrt{A/V}$ . So  $K_p$  with subthreshold conduction can be found by extrapolating of the high  $V_{gs}$  slope back to the threshold voltage intercept then doubling the increase. We need to know  $V_{to}$  approximately; a first approximation can be obtained from a  $\sqrt{I_d}$  versus  $V_{gs}$  plot x-axis intercept (plus about 30-50mV) OR use the  $V_{gs}$  where  $d\sqrt{I_d}$  is about 60% of the plots peak (see Fig A4.1 arrow at  $0.42\sqrt{A/V}$ ).

It is also possible to extract the subthreshold slope from the rising edge slope of the  $d\sqrt{I_d}$   $R_s$  can be extracted from the high  $V_{gs}$  slope.

